// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/adapters/UniswapV3Adapter.sol



// SPDX-License-Identifier: MIT OR Apache-2.0
pragma solidity ^0.8.17;

import {Ownable} from "@openzeppelin/contracts/access/Ownable.sol";
import {IERC20} from "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import {ISwapRouter} from "../interfaces/ISwapRouter.sol";

/// @title UniswapV3Adapter
/// @author Nocturne Labs
/// @notice Adapter contract for interacting with the Uniswap SwapRouter contract. Technically
///         Nocturne does not need an adapter but for the sake of avoiding attacks that bypass
///         deposit limits, we needed some calldata validation. This contract validates the tokenIn
///         of a single swap and path of a multi-hop swap to ensure no custom-deployed tokens are
///         used.
/// @dev This contract is Ownable and non-upgradeable. The only power the owner has is adding
///      and removing tokens from the whitelist.
contract UniswapV3Adapter is Ownable {
    // UniswapV3 SwapRouter
    ISwapRouter public immutable _swapRouter;

    // Whitelist of tokens that can be used in a path
    mapping(address => bool) public _allowedTokens;

    /// @notice Event emitted when a token is added to the whitelist
    event TokenPermissionSet(address token, bool permission);

    // Constructor, sets uniswap swap router contract
    constructor(address swapRouter) Ownable() {
        _swapRouter = ISwapRouter(swapRouter);
    }

    /// @notice Set token permission in whitelist
    /// @param token Token address
    /// @param permission Whether token is allowed
    function setTokenPermission(
        address token,
        bool permission
    ) external onlyOwner {
        _allowedTokens[token] = permission;
        emit TokenPermissionSet(token, permission);
    }

    /// @notice Proxy for exactInputSingle function of Uniswap SwapRouter. Ensures tokenIn and
    ///         tokenOut are allowed.
    /// @param params ExactInputSingleParams (see ISwapRouter)
    function exactInputSingle(
        ISwapRouter.ExactInputSingleParams memory params
    ) external returns (uint256 amountOut) {
        require(_allowedTokens[params.tokenIn], "!tokenIn");
        require(_allowedTokens[params.tokenOut], "!tokenOut");

        IERC20(params.tokenIn).transferFrom(
            address(msg.sender),
            address(this),
            params.amountIn
        );
        IERC20(params.tokenIn).approve(address(_swapRouter), params.amountIn);

        amountOut = _swapRouter.exactInputSingle(params);
    }

    /// @notice Proxy for exactInput function of Uniswap SwapRouter.
    ///         Ensures all tokens in path are is allowed.
    /// @param params ExactInputParams (see ISwapRouter)
    function exactInput(
        ISwapRouter.ExactInputParams memory params
    ) external returns (uint256 amountOut) {
        require(tokensAreAllowed(params.path), "!allowed path");

        address tokenIn = extractTokenAddressFromPath(params.path, 0);

        IERC20(tokenIn).transferFrom(
            address(msg.sender),
            address(this),
            params.amountIn
        );
        IERC20(tokenIn).approve(address(_swapRouter), params.amountIn);

        amountOut = _swapRouter.exactInput(params);
    }

    /// @notice Checks that all tokens in path are allowed, returns false if any are not
    /// @param path Path of tokens
    /// @dev One Uniswap pool is denoted as (address tokenIn, uint24 poolFee, address tokenOut).
    ///      The path param describes which pools are used in the swap. Path is encoded as follows:
    ///      `(address tokenIn) || hops`, where `hops` is a sequence of `(uint24 fee, address
    ///      tokenOut)` pairs (i.e. for an intermediate hop, the tokenOut of the last hop is the
    ///      tokenIn of the next). The length of path will always be `20 + ((20 + 3) * N)` bytes.
    function tokensAreAllowed(bytes memory path) internal view returns (bool) {
        // NOTE: function reverts if path is not the correct length 20 + ((20 + 3) * n)
        uint256 numTokens = mustGetNumTokensInPath(path);

        for (uint256 i = 0; i < numTokens; i++) {
            address tokenAddress = extractTokenAddressFromPath(path, i);
            if (!_allowedTokens[tokenAddress]) {
                return false;
            }
        }

        return true;
    }

    function mustGetNumTokensInPath(
        bytes memory path
    ) internal pure returns (uint256) {
        uint256 length = path.length;

        // Ensure the path length is correct: ((20 + 3) * n) + 20
        require((length - 20) % 23 == 0, "Invalid path length");

        return (length - 20) / 23 + 1;
    }

    /// @notice Extracts token address from path at a given index into the path
    /// @param path Path of tokens
    /// @param tokenIndex Index of token in path, 0 for tokenIn, N for tokenOut for N hop path
    function extractTokenAddressFromPath(
        bytes memory path,
        uint256 tokenIndex
    ) internal pure returns (address) {
        // Load 32 bytes that are at offset 0 + 32 (length prefix) + index * 23 bytes
        // Shift right by 12 bytes to get the token address, which is first 20 bytes
        address tokenAddr;
        assembly {
            tokenAddr := shr(
                96,
                mload(add(add(path, 0x20), mul(tokenIndex, 0x17)))
            )
        }

        return tokenAddr;
    }
}

// File: contracts/interfaces/ISwapRouter.sol



// SPDX-License-Identifier: MIT OR Apache-2.0
pragma solidity ^0.8.17;

// Forked/modified from: https://github.com/Uniswap/v3-periphery/blob/main/contracts/interfaces/ISwapRouter.sol

/// @title Router token swapping functionality
/// @notice Functions for swapping tokens via Uniswap V3
interface ISwapRouter {
    struct ExactInputSingleParams {
        address tokenIn;
        address tokenOut;
        uint24 fee;
        address recipient;
        uint256 deadline;
        uint256 amountIn;
        uint256 amountOutMinimum;
        uint160 sqrtPriceLimitX96;
    }

    /// @notice Swaps `amountIn` of one token for as much as possible of another token
    /// @param params The parameters necessary for the swap, encoded as `ExactInputSingleParams` in calldata
    /// @return amountOut The amount of the received token
    function exactInputSingle(
        ExactInputSingleParams calldata params
    ) external payable returns (uint256 amountOut);

    struct ExactInputParams {
        bytes path;
        address recipient;
        uint256 deadline;
        uint256 amountIn;
        uint256 amountOutMinimum;
    }

    /// @notice Swaps `amountIn` of one token for as much as possible of another along the specified path
    /// @param params The parameters necessary for the multi-hop swap, encoded as `ExactInputParams` in calldata
    /// @return amountOut The amount of the received token
    function exactInput(
        ExactInputParams calldata params
    ) external payable returns (uint256 amountOut);

    struct ExactOutputSingleParams {
        address tokenIn;
        address tokenOut;
        uint24 fee;
        address recipient;
        uint256 deadline;
        uint256 amountOut;
        uint256 amountInMaximum;
        uint160 sqrtPriceLimitX96;
    }

    /// @notice Swaps as little as possible of one token for `amountOut` of another token
    /// @param params The parameters necessary for the swap, encoded as `ExactOutputSingleParams` in calldata
    /// @return amountIn The amount of the input token
    function exactOutputSingle(
        ExactOutputSingleParams calldata params
    ) external payable returns (uint256 amountIn);

    struct ExactOutputParams {
        bytes path;
        address recipient;
        uint256 deadline;
        uint256 amountOut;
        uint256 amountInMaximum;
    }

    /// @notice Swaps as little as possible of one token for `amountOut` of another along the specified path (reversed)
    /// @param params The parameters necessary for the multi-hop swap, encoded as `ExactOutputParams` in calldata
    /// @return amountIn The amount of the input token
    function exactOutput(
        ExactOutputParams calldata params
    ) external payable returns (uint256 amountIn);
}